


# Deflection of a Beam with Distributed and Point Load

## ▼ Introduction

This application will derive an explicit expression for the deflection of a beam with a distributed load and a point load.



## ▼ Governing Equations

> *restart* :

The [Euler-Bernoulli](#) equation

$$> de := EI \cdot \frac{d^4}{dx^4} w(x) = q(x) :$$

Initial and boundary conditions

$$> ibc := w(0) = 0, w(L) = 0, (D@@2)(w)(0) = 0, (D@@2)(w)(L) = 0 :$$

Distributed load and point load

$$> q := x \rightarrow Q \cdot (1 - \text{Heaviside}(x - a)) + F \cdot \text{Dirac}(x - b) :$$

## ▼ Solution of the Differential Equation

Solve the differential equation together with the initial/boundary conditions and the load distribution to get an explicit expression for the beam deflection.

$$> deSol := dsolve(\{de, ibc\}, w(x)) : \\ deflection := simplify(rhs(deSol), symbolic)$$

$$deflection := \frac{1}{144 EI L} \left( -Q x (L - x) (L + x) (L - a)^4 \text{Dirac}(1, L - a) + 4 F x (L - x) (L \quad (3.1)$$

$$+ x) (L - b)^3 \text{Dirac}(1, L - b) - 8 Q x (L - x) (L + x) (L - a)^3 \text{Dirac}(L - a)$$

$$- 6 Q x (L - a)^2 (L^2 + 2 L a - a^2 - 2 x^2) \text{Heaviside}(L - a) + 24 F x (L - x) (L$$

$$+ x) (L - b)^2 \text{Dirac}(L - b) + 48 x F \left( -\frac{x^2}{2} + b \left( L - \frac{b}{2} \right) \right) (L - b) \text{Heaviside}(L - b)$$

$$- 6 L Q (-x + a)^4 \text{Heaviside}(x - a) - 24 F L (-x + b)^3 \text{Heaviside}(x - b)$$

$$- 48 \left( \frac{a^2 Q \left( L x - \frac{1}{4} a^2 - \frac{1}{2} x^2 \right) \text{Heaviside}(-a)}{2} + F b \left( L x - \frac{1}{2} b^2 \right.$$

$$\left. - \frac{1}{2} x^2 \right) \text{Heaviside}(-b) - \frac{Q x L (L^2 + L x - x^2)}{8} \right) (L - x)$$

Derive the moment and shear distribution.

$$> moment := EI \cdot \text{diff}(deflection, x, x)$$

$$moment := \frac{1}{144 L} \left( 2 Q x (L - a)^4 \text{Dirac}(1, L - a) - 2 Q (L - x) (L - a)^4 \text{Dirac}(1, L - a) \quad (3.2)$$

$$- 144 F L (-x + b) \text{Heaviside}(x - b) - 24 F L (-x + b)^3 \text{Dirac}(1, x - b) + 48 a^2 Q (L$$

$$- x) \text{Heaviside}(-a) + 96 F b (L - x) \text{Heaviside}(-b) - 12 Q L (L^2 + L x - x^2)$$

$$- 12 Q x L (L - 2 x) + 2 Q (L + x) (L - a)^4 \text{Dirac}(1, L - a) - 8 F (L + x) (L$$

$$\begin{aligned}
& -b)^3 \text{Dirac}(1, L-b) + 16 Q (L+x) (L-a)^3 \text{Dirac}(L-a) + 72 Q (L \\
& -a)^2 x \text{Heaviside}(L-a) - 48 F (L+x) (L-b)^2 \text{Dirac}(L-b) - 144 F x (L \\
& -b) \text{Heaviside}(L-b) + 48 L Q (-x+a)^3 \text{Dirac}(x-a) - 72 L Q (-x \\
& +a)^2 \text{Heaviside}(x-a) - 6 L Q (-x+a)^4 \text{Dirac}(1, x-a) + 144 F L (-x \\
& +b)^2 \text{Dirac}(x-b) - 48 \left( -\frac{Q a^2 \text{Heaviside}(-a)}{2} - F b \text{Heaviside}(-b) \right. \\
& \left. - \frac{Q L (L-2x)}{4} + \frac{Q x L}{4} \right) (L-x) + 8 F (L-x) (L-b)^3 \text{Dirac}(1, L-b) \\
& - 8 F x (L-b)^3 \text{Dirac}(1, L-b) - 16 Q (L-x) (L-a)^3 \text{Dirac}(L-a) \\
& + 16 Q x (L-a)^3 \text{Dirac}(L-a) + 48 F (L-x) (L-b)^2 \text{Dirac}(L-b) - 48 F x (L \\
& -b)^2 \text{Dirac}(L-b) )
\end{aligned}$$

> *shear* := *diff*(*moment*, *x*)

$$\begin{aligned}
\text{shear} := \frac{1}{144 L} & ( -144 F b \text{Heaviside}(-b) - 72 Q a^2 \text{Heaviside}(-a) + 6 Q (L \\
& -a)^4 \text{Dirac}(1, L-a) + 144 F L \text{Heaviside}(x-b) - 36 Q L (L-x) - 36 Q L (L \\
& -2x) - 24 F (L-b)^3 \text{Dirac}(1, L-b) + 48 Q (L-a)^3 \text{Dirac}(L-a) + 72 Q (L \\
& -a)^2 \text{Heaviside}(L-a) - 144 F (L-b)^2 \text{Dirac}(L-b) - 144 F (L-b) \text{Heaviside}(L \\
& -b) + 72 L Q (-x+a)^3 \text{Dirac}(1, x-a) - 216 L Q (-x+a)^2 \text{Dirac}(x-a) \\
& + 144 L Q (-x+a) \text{Heaviside}(x-a) - 6 L Q (-x+a)^4 \text{Dirac}(2, x-a) - 432 F L ( \\
& -x+b) \text{Dirac}(x-b) - 24 F L (-x+b)^3 \text{Dirac}(2, x-b) + 216 F L (-x \\
& +b)^2 \text{Dirac}(1, x-b) + 36 Q x L )
\end{aligned} \tag{3.3}$$

## ▼ Plot the Deflection, Moment, and Shear

Assign parameters.


```

> Q := -12 :
  F := -10 :
  L := 20 :
  a := 3 :
  b := 5 :
  EI := 10000 :

```

Plot deflection, moment, and shear.

```
> plot(deflection, x = 0 ..L, size = [ 1000, 400 ], axesfont = [ Calibri ], title = "Deflection", labels
      = [ "Distance along beam", "Deflection" ], labeldirections = [ horizontal, vertical ], labelfont
      = [ Calibri ], titlefont = [ Calibri, 16, bold ], background = ColorTools:-Color("RGB", [ 218
      / 255, 223 / 255, 225 / 255 ]), axis = [ gridlines = [ color = ColorTools:-Color("RGB", [ 1, 1,
      1 ]) ]])
```




```
> plot(moment, x = 0 ..L, size = [ 1000, 400 ], axesfont = [ Calibri ], title = "Moment", labels
      = [ "Distance along beam", "Moment" ], labeldirections = [ horizontal, vertical ], labelfont
      = [ Calibri ], titlefont = [ Calibri, 16, bold ], background = ColorTools:-Color("RGB", [ 218
      / 255, 223 / 255, 225 / 255 ]), axis = [ gridlines = [ color = ColorTools:-Color("RGB", [ 1, 1,
      1 ]) ]])
```



### Shear Distribution

```
> plot(shear, x = 0..L, size = [1000, 400], axesfont = [Calibri], title = "Shear", labels
      = ["Distance along beam", "Shear"], labeldirections = [horizontal, vertical], labelfont
      = [Calibri], titlefont = [Calibri, 16, bold], background = ColorTools:-Color("RGB", [218
      /255, 223/255, 225/255]), axis = [gridlines = [color = ColorTools:-Color("RGB", [1, 1,
      1])]])
```

# Shear



>