


# Theoretical Performance of a LOX-LH2 Rocket

## ▼ Introduction

Liquid hydrogen (at 20.27 K) and liquid oxygen (at 90.17 K) are burned in the combustion chamber (with an infinite area ratio) of a rocket.



The combustion products contain  $H_2$ ,  $O_2$ , H, O, OH,  $HO_2$  and  $H_2O_2$

Chemical	In Feed	In Products
$H_2$	1	n1
$O_2$	0.34974	n2
$H_2O$		n3

O		n4
H		n5
OH		n6
HO <sub>2</sub>		n7
H <sub>2</sub> O <sub>2</sub>		n8

This application will calculate

- the adiabatic flame temperature and composition of the combustion products
- the pressures and temperatures in the throat and exit
- and the theoretical rocket performance, including the ideal specific impulse, characteristic velocity, sonic velocity and more

Assumptions

- The combustion chamber is large compared to the throat, hence the assumption of an infinite area ratio
- The flow composition does not change through the nozzle expansion (i.e. reaction rate is slow compared to flowrate). This is also known as "frozen" flow

## ▼ Physical Properties

```
> restart:
> with(ThermophysicalData:-Chemicals):
```

Enthalpies

```
> h_O2 := Property("Hmolar", "O2", "temperature" = T):
h_H2 := Property("Hmolar", "H2", "temperature" = T):
h_H2O := Property("Hmolar", "H2O", "temperature" = T):
h_OH := Property("Hmolar", "OH", "temperature" = T):
h_H := Property("Hmolar", "H", "temperature" = T):
h_O := Property("Hmolar", "O", "temperature" = T):
h_HO2 := Property("Hmolar", "HO2", "temperature" = T):
h_H2O2 := Property("Hmolar", "H2O2", "temperature" = T):
```

Entropies

```
> s_O2 := Property("Smolar", "O2", "temperature" = T) - R *
ln(P_c/P_s):
s_H2 := Property("Smolar", "H2", "temperature" = T) - R *
ln(P_c/P_s):
s_H2O := Property("Smolar", "H2O", "temperature" = T) - R *
ln(P_c/P_s):
s_OH := Property("Smolar", "OH", "temperature" = T) - R *
ln(P_c/P_s):
s_H := Property("Smolar", "H", "temperature" = T) - R *
```

```

ln(P_c/P_s):
s_O := Property("Smolar", "O", "temperature" = T) - R *
ln(P_c/P_s):
s_HO2 := Property("Smolar", "HO2", "temperature" = T) - R *
ln(P_c/P_s):
s_H2O2 := Property("Smolar", "H2O2", "temperature" = T) - R *
ln(P_c/P_s):

```

Enthalpy of formation

```

> h_f_O2L := Property("HeatOfFormation", "O2 (L)");
h_f_H2L := Property("HeatOfFormation", "H2 (L)");
h_f_O2 := Property("HeatOfFormation", "O2");
h_f_H2 := Property("HeatOfFormation", "H2");
h_f_H2O := Property("HeatOfFormation", "H2O");
h_f_OH := Property("HeatOfFormation", "OH");
h_f_H := Property("HeatOfFormation", "H");
h_f_O := Property("HeatOfFormation", "O");
h_f_HO2 := Property("HeatOfFormation", "HO2");
h_f_H2O2 := Property("HeatOfFormation", "H2O2");

h_f_O2L := -12979.000
h_f_H2L := -9012.000
h_f_O2 := 0.
h_f_H2 := 0.
h_f_H2O := -241826.000
h_f_OH := 37278.206
h_f_H := 217998.828
h_f_O := 249175.003
h_f_HO2 := 12020.000
h_f_H2O2 := -135880.000

```

(2.1)

Reference enthalpies

```

> h_r_O2 := eval(h_O2, T = 298.15);
h_r_H2 := eval(h_H2, T = 298.15);
h_r_H2O := eval(h_H2O, T = 298.15);
h_r_OH := eval(h_OH, T = 298.15);
h_r_H := eval(h_H, T = 298.15);
h_r_O := eval(h_O, T = 298.15);
h_r_HO2 := eval(h_HO2, T = 298.15);
h_r_H2O2 := eval(h_H2O2, T = 298.15)

h_r_O2 := 0.
h_r_H2 := -4.957942313 10-6
h_r_H2O := -241826.0005
h_r_OH := 37278.20600
h_r_H := 217998.8279
h_r_O := 249175.0027
h_r_HO2 := 12019.99997
h_r_H2O2 := -135880.0000

```

(2.2)

## Gibbs Free Energy of Formation of the combustion products

```
> Gibbs_H2 := proc(temp)
    return 0
end proc:

> Gibbs_O2 := proc(temp)
    return 0
end proc:

> Gibbs_H2O := proc(temp)
    local DeltaH, DeltaS, DeltaG:
    DeltaH := eval(h_H2O - (h_H2 + 0.5*h_O2), T = temp):
    DeltaS := eval(s_H2O - (s_H2 + 0.5*s_O2), T = temp):
    DeltaG := DeltaH - DeltaS*temp:
end proc:

> Gibbs_O := proc(temp)
    local DeltaH, DeltaS, DeltaG:
    DeltaH := eval(h_O - 0.5*h_O2, T = temp):
    DeltaS := eval(s_O - 0.5*s_O2, T = temp):
    DeltaG := DeltaH - DeltaS*temp:
    return DeltaG:
end proc:

> Gibbs_H := proc(temp)
    local DeltaH, DeltaS, DeltaG:
    DeltaH := eval(h_H - 0.5*h_H2, T = temp):
    DeltaS := eval(s_H - 0.5*s_H2, T = temp):
    DeltaG := DeltaH - DeltaS*temp:
    return DeltaG:
end proc:

> Gibbs_OH := proc(temp)
    local DeltaH, DeltaS, DeltaG:
    DeltaH := eval(h_OH - (0.5*h_H2 + 0.5*h_O2), T = temp):
    DeltaS := eval(s_OH - (0.5*s_H2 + 0.5*s_O2), T = temp):
    DeltaG := DeltaH - DeltaS*temp:
    return DeltaG:
end proc:

> Gibbs_HO2 := proc(temp)
    local DeltaH, DeltaS, DeltaG:
    DeltaH := eval(h_HO2 - (0.5*h_H2 + h_O2), T = temp):
    DeltaS := eval(s_HO2 - (0.5*s_H2 + s_O2), T = temp):
    DeltaG := DeltaH - DeltaS*temp:
    return DeltaG:
end proc:

Gibbs_H2O2 := proc(temp)
    local DeltaH, DeltaS, DeltaG:
    DeltaH := eval(h_H2O2 - (h_H2 + h_O2), T = temp):
    DeltaS := eval(s_H2O2 - (s_H2 + s_O2), T = temp):
    DeltaG := DeltaH - DeltaS*temp:
    return DeltaG:
end proc:
```

Ideal gas constant

```
> R := 8.3144:
```

Chamber Pressure

```
> P_c := 53.3172 * Unit(bar):
```

Atmospheric Pressure

```
> P_a := 1.01325 * Unit(bar):
```

Standard pressure

```
> P_s := 1.0 * Unit(bar):
```

Ratio of nozzle exit and combustion chamber throat area

```
> epsilon := 1.0:
```

Ratio of exit area to throat area

```
> AeAt := 1.58:
```

Moles of H<sub>2</sub> and O<sub>2</sub> fed into combustion chamber (only the ratio is important)


```
> N_H2L := 1:
  N_O2L := 0.34974:
```

Mach number at throat (= 1 for choked flow)

```
> M_t := 1:
```

## ▼ Equilibrium Composition

Equating the moles of chemicals in the reactants and products gives



Balance on H atoms

$$2 n_1 + 2 n_3 + n_4 + n_6 + n_7 + 2 n_8 = 2 * N_{H2L}$$

Balance on O atoms

$$2 n_2 + n_3 + n_5 + n_6 + 2 n_7 + 2 n_8 = 2 * N_{O2L}$$

Hence the constraints are

```
> con1 := 2 * n1 + 2 * n3 + n4 + n6 + n7 + 2 * n8 = 2 * N_H2L:
> con2 := 2 * n2 + n3 + n5 + n6 + 2 * n7 + 2 * n8 = 2 * N_O2L:
```

Total moles of products

```
> nt := n1 + n2 + n3 + n4 + n5 + n6 + n7 + n8:
```

For a given temperature, minimizing the Gibbs Free Energy of the combustion products will give the equilibrium composition

```
> gibbs := n1 * (Gibbs_H2(T) + R * T * ln(n1/nt))
+ n2 * (Gibbs_O2(T) + R * T * ln(n2/nt))
+ n3 * (Gibbs_H2O(T) + R * T * ln(n3/nt))
+ n4 * (Gibbs_H(T) + R * T * ln(n4/nt))
+ n5 * (Gibbs_O(T) + R * T * ln(n5/nt))
+ n6 * (Gibbs_OH(T) + R * T * ln(n6/nt))
+ n7 * (Gibbs_HO2(T) + R * T * ln(n7/nt))
+ n8 * (Gibbs_H2O2(T) + R * T * ln(n8/nt)) :
```

Hence the values of  $n_1$ ,  $n_2$ ,  $n_3$ ,  $n_4$ ,  $n_5$ ,  $n_6$ ,  $n_7$  and  $n_8$  are given by the numeric solution of these equations, where  $L_1$  and  $L_2$  are the Lagrange multipliers.

```
> eqComposition := L1 * diff(lhs(con1), n1) + L2 * diff(lhs(con2),
n1) = diff(gibbs, n1),
L1 * diff(lhs(con1), n2) + L2 * diff(lhs(con2),
n2) = diff(gibbs, n2),
L1 * diff(lhs(con1), n3) + L2 * diff(lhs(con2),
n3) = diff(gibbs, n3),
L1 * diff(lhs(con1), n4) + L2 * diff(lhs(con2),
n4) = diff(gibbs, n4),
L1 * diff(lhs(con1), n5) + L2 * diff(lhs(con2),
n5) = diff(gibbs, n5),
L1 * diff(lhs(con1), n6) + L2 * diff(lhs(con2),
n6) = diff(gibbs, n6),
L1 * diff(lhs(con1), n7) + L2 * diff(lhs(con2),
n7) = diff(gibbs, n7),
L1 * diff(lhs(con1), n8) + L2 * diff(lhs(con2),
n8) = diff(gibbs, n8) :
```

## ▼ Heat Balance

The flame temperature is given by equating the heat of the reactants to the heat of the products

```
> H_reactants := N_H2L * h_f_H2L + N_O2L * h_f_O2L;
-13.55 × 103 (4.1)
```

```
> H_products := + n1 * (h_f_H2 + (h_H2 - h_r_H2))
+ n2 * (h_f_O2 + (h_O2 - h_r_O2))
+ n3 * (h_f_H2O + (h_H2O - h_r_H2O))
+ n4 * (h_f_H + (h_H - h_r_H))
+ n5 * (h_f_O + (h_O - h_r_O))
+ n6 * (h_f_OH + (h_OH - h_r_OH))
+ n7 * (h_f_HO2 + (h_HO2 - h_r_HO2))
+ n8 * (h_f_H2O2 + (h_H2O2 - h_r_H2O2)) :
```

```
> flameTemp := H_reactants = H_products:
```

## ▼ Numerical Solution of Equilibrium Composition and Flame Temperature

```
> res:=fsolve({eqComposition, flameTemp, con1, con2}, {L1 =
-1000, L2=-1000, T = 3000, n1 = 0.1, n2 = 0.1, n2 = 0.1, n3 =
```

$$\begin{aligned}
 &0.1, n4 = 0.1, n5 = 0.1, n6 = 0.1, n7 = 0.1, n8 = 0.1\}) \\
 \text{res} := \{ &L1 = -17182.81432, L2 = -89531.03079, T = 3383.844425, n1 = 0.3061915322, n2 \\
 &= 0.001788202357, n3 = 0.6590840027, n4 = 0.03479245713, n5 = 0.002147607467, n6 \\
 &= 0.03462930001, n7 = 0.00001551202423, n8 = 5.830551718 \times 10^{-6} \}
 \end{aligned} \tag{5.1}$$

Hence the temperature in the rocket combustion chamber is

$$\begin{aligned}
 &> T\_c := \text{eval}(T, \text{res}) * \text{Unit}(K) \\
 &T\_c := 3383.844425 K
 \end{aligned} \tag{5.2}$$

The equilibrium composition of the combustion products are (in moles)

$$\begin{aligned}
 &> \text{mol\_H2} := \text{eval}(n1, \text{res}); \\
 &\text{mol\_O2} := \text{eval}(n2, \text{res}); \\
 &\text{mol\_H2O} := \text{eval}(n3, \text{res}); \\
 &\text{mol\_H} := \text{eval}(n4, \text{res}); \\
 &\text{mol\_O} := \text{eval}(n5, \text{res}); \\
 &\text{mol\_OH} := \text{eval}(n6, \text{res}); \\
 &\text{mol\_HO2} := \text{eval}(n7, \text{res}); \\
 &\text{mol\_H2O2} := \text{eval}(n8, \text{res}); \\
 &306.19 \times 10^{-3} \\
 &1.79 \times 10^{-3} \\
 &659.08 \times 10^{-3} \\
 &34.79 \times 10^{-3} \\
 &2.15 \times 10^{-3} \\
 &34.63 \times 10^{-3} \\
 &15.51 \times 10^{-6} \\
 &5.83 \times 10^{-6}
 \end{aligned} \tag{5.3}$$

$$\begin{aligned}
 &> \text{mol\_total} := \text{mol\_H2} + \text{mol\_H2O} + \text{mol\_O2} + \text{mol\_H} + \text{mol\_O} + \text{mol\_OH} \\
 &\quad + \text{mol\_HO2} + \text{mol\_H2O2} \\
 &\text{mol\_total} := 1.038654444
 \end{aligned} \tag{5.4}$$

Mole fractions in the combustion products

$$\begin{aligned}
 &> \text{molFrac\_H2} := \text{mol\_H2}/\text{mol\_total}; \\
 &\text{molFrac\_O2} := \text{mol\_O2}/\text{mol\_total}; \\
 &\text{molFrac\_H2O} := \text{mol\_H2O}/\text{mol\_total}; \\
 &\text{molFrac\_H} := \text{mol\_H}/\text{mol\_total}; \\
 &\text{molFrac\_O} := \text{mol\_O}/\text{mol\_total}; \\
 &\text{molFrac\_OH} := \text{mol\_OH}/\text{mol\_total}; \\
 &\text{molFrac\_HO2} := \text{mol\_HO2}/\text{mol\_total}; \\
 &\text{molFrac\_H2O2} := \text{mol\_H2O2}/\text{mol\_total}; \\
 &\text{molFrac\_H2} := 0.2947963435 \\
 &\text{molFrac\_O2} := 0.001721652824 \\
 &\text{molFrac\_H2O} := 0.6345556085 \\
 &\text{molFrac\_H} := 0.03349762506 \\
 &\text{molFrac\_O} := 0.002067682355 \\
 &\text{molFrac\_OH} := 0.03334053997
 \end{aligned}$$

$$\begin{aligned} \text{molFrac\_HO2} &:= 0.00001493473053 \\ \text{molFrac\_H2O2} &:= 5.61356257810^{-6} \end{aligned}$$

(5.5)

## ▼ Ideal Performance of an Infinite Area Ratio Rocket

```
> with(Units[Simple]):
```

Ideal gas constant

```
> R := 8.3144 * Unit(J/mol/K):
```

Gravity

```
> grav := 9.81*Unit(m/s^2):
```

Molecular weight of the combustion products

```
> Mw_mix := molFrac_H2 * Property("MolarMass", "H2",
useunits)
+ molFrac_H2O * Property("MolarMass", "H2O",
useunits)
+ molFrac_O2 * Property("MolarMass", "O2",
useunits)
+ molFrac_H * Property("MolarMass", "H",
useunits)
+ molFrac_O * Property("MolarMass", "O",
useunits)
+ molFrac_OH * Property("MolarMass", "OH",
useunits)
+ molFrac_HO2 * Property("MolarMass", "HO2",
useunits)
+ molFrac_H2O2 * Property("MolarMass", "H2O2",
useunits)
```

$$12.72 \frac{\text{g}}{\text{mol}}$$

(6.1)

Specific heat capacity (at constant pressure) in the combustion chamber

```
> Cp_c_mol := molFrac_H2 * Property("Cpmolar", "H2",
"temperature" = T_c)
+ molFrac_H2O * Property("Cpmolar", "H2O",
"temperature" = T_c)
+ molFrac_O2 * Property("Cpmolar", "O2",
"temperature" = T_c)
+ molFrac_O * Property("Cpmolar", "O",
"temperature" = T_c)
+ molFrac_H * Property("Cpmolar", "H",
"temperature" = T_c)
+ molFrac_OH * Property("Cpmolar", "OH",
"temperature" = T_c)
+ molFrac_HO2 * Property("Cpmolar", "HO2",
"temperature" = T_c)
+ molFrac_H2O2 * Property("Cpmolar", "H2O2",
"temperature" = T_c)
```

$$50.02 \frac{\text{J}}{\text{mol K}} \quad (6.2)$$

Specific heat capacity (at constant volume) in the combustion chamber

> Cv\_c\_mol := Cp\_c\_mol - R

$$41.70 \frac{\text{J}}{\text{mol K}} \quad (6.3)$$

Isentropic expansion coefficient in the chamber

> Gamma\_c := Cp\_c\_mol / Cv\_c\_mol

$$1.199 \quad (6.4)$$

Mach number at exit

> M\_e := fsolve(AeAt = ((Gamma\_c + 1)/2) ^ (-(Gamma\_c + 1)/(2 \* (Gamma\_c - 1))) \* (1 + 0.5 \* (Gamma\_c - 1) \* M\_e^2) ^ ((Gamma\_c + 1)/(2 \* (Gamma\_c - 1))) / M\_e, M\_e = 1)

$$.41 \quad (6.5)$$

Throat temperature

> T\_t := T\_c \* (1 + (Gamma\_c - 1)/2 \* M\_t^2) ^ (-1)

$$3077.11 \text{ K} \quad (6.6)$$

Exit temperature

> T\_e := T\_c \* (1 + (Gamma\_c - 1)/2 \* M\_e^2) ^ (-1)

$$3327.85 \text{ K} \quad (6.7)$$

Specific heat capacity (at constant pressure) at throat

> Cp\_t\_mol := molFrac\_H2 \* Property("Cpmolar", "H2",  
"temperature" = T\_t)  
+ molFrac\_H2O \* Property("Cpmolar", "H2O",  
"temperature" = T\_t)  
+ molFrac\_O2 \* Property("Cpmolar", "O2",  
"temperature" = T\_t)  
+ molFrac\_H \* Property("Cpmolar", "H",  
"temperature" = T\_t)  
+ molFrac\_O \* Property("Cpmolar", "O",  
"temperature" = T\_t)  
+ molFrac\_OH \* Property("Cpmolar", "OH",  
"temperature" = T\_t)  
+ molFrac\_HO2 \* Property("Cpmolar", "HO2",  
"temperature" = T\_t)  
+ molFrac\_H2O2 \* Property("Cpmolar", "H2O2",  
"temperature" = T\_t)

$$49.25 \frac{\text{J}}{\text{mol K}} \quad (6.8)$$

Isentropic expansion coefficient at throat

> Gamma\_t := Cp\_t\_mol / (Cp\_t\_mol - R)

$$1.203 \quad (6.9)$$

Throat pressure

> P\_t := P\_c \* (1 + (Gamma\_c - 1)/2 \* M\_t^2) ^ (Gamma\_c / (1 -

$$\text{Gamma\_c}) )$$

$$30.10 \text{ bar} \quad (6.10)$$

Exit pressure

$$> P_e := P_c * (1 + (\text{Gamma}_t - 1)/2 * M_e^2)^{(\text{Gamma}_t / (1 - \text{Gamma}_t))}$$

$$48.21 \text{ bar} \quad (6.11)$$

Chamber gas density

$$> \rho_c := P_c * M_{w\_mix} / (R * T_c)$$

$$2.41 \frac{\text{kg}}{\text{m}^3} \quad (6.12)$$

Throat gas density

$$> \rho_t := P_t * M_{w\_mix} / (R * T_t)$$

$$1.50 \frac{\text{kg}}{\text{m}^3} \quad (6.13)$$

Sonic velocity in chamber and throat

$$> \text{sonicVelocity}_c := \sqrt{(\text{Gamma}_c * P_c / \rho_c)}$$

$$1629.02 \frac{\text{m}}{\text{s}} \quad (6.14)$$

$$> \text{sonicVelocity}_t := \sqrt{(\text{Gamma}_t * P_t / \rho_t)}$$

$$1555.86 \frac{\text{m}}{\text{s}} \quad (6.15)$$

Throat velocity for an isentropic nozzle

$$> V_t := \sqrt{\frac{2 \cdot T_c R}{M_{w\_mix}} \cdot \frac{\text{Gamma}_c}{\text{Gamma}_c - 1} \left( 1 - \left( \frac{P_t}{P_c} \right)^{\frac{\text{Gamma}_c - 1}{\text{Gamma}_c}} \right)}$$

$$1553.44 \frac{\text{m}}{\text{s}} \quad (6.16)$$

Ideal specific impulse

$$> \text{Isp}_{ideal} := V_t / \text{grav}$$

$$158.35 \text{ s} \quad (6.17)$$

Ideal specific impulse as defined by NASA CEA

$$> \text{Isp}_{ideal\_NASA} := V_t$$

$$1553.44 \frac{\text{m}}{\text{s}} \quad (6.18)$$

Ideal specific impulse in a vacuum.

$$> \text{Isp}_{vac} := V_t + P_t / (\rho_t * V_t)$$

$$2848.65 \frac{\text{m}}{\text{s}} \quad (6.19)$$

Characteristic velocity (C-Star)

$$> C_{star} := \sqrt{1 / \text{Gamma}_c * ((\text{Gamma}_c + 1)/2) ^ ((\text{Gamma}_c + 1) / (\text{Gamma}_c - 1)) * R / M_{w\_mix} * T_c}$$

$$2294.05 \frac{\text{m}}{\text{s}}$$

**(6.20)**

Legal Notice: © Maplesoft, a division of Waterloo Maple Inc. 2018. Maplesoft and Maple are trademarks of Waterloo Maple Inc. This application may contain errors and Maplesoft is not liable for any damages resulting from the use of this material. This application is intended for non-commercial, non-profit use only. Contact Maplesoft for permission if you wish to use this application in for-profit activities.